

Simulation, debugging and implementing
the design with
ModelSim
and
Quartus Prime Lite

Contents

Introduction	3
Prerequisites	3
Objectives.....	3
Before going forward	3
Creating the design in Quartus Prime	4
Design overview	4
Creating the project in Quartus Prime	5
Simulating the design in ModelSim.....	6
Top-level file for simulation	6
Launching ModelSim	8
Simulation and debugging.....	8
Debugging the design.....	11
Add IP components for debugging.....	11
Top-level file for debugging	16
Analysis and Elaboration	16
Adding and setting up Signal TapII	17
Full compilation.....	22
Debugging the design.....	23
Implementing the design	30
Top-level file for implementation	30
Full compilation.....	31
Programming and verifying the board	32
Conclusions	33

Introduction

Traditional debugging techniques often involve using an external pattern generator to exercise the logic and a logic analyzer to study the output waveforms during runtime. The In-System Sources and Probes Editor (ISSP) in the Quartus software extends the portfolio of verification tools. It allows you to monitor various signals in the design.

ISSP Editor consists of a probe function and interface to control the instances during run time. It operates over JTAG. Each instance of ISSP can drive and toggle values up to 512 signals. It can create up to 128 instances of ISSP using IP Catalog.

The main difference between ISSP and Signal Tap (discussed in the next section) is that ISSP does not have a clock as reference to the output signals we will probe, it will only sample signals in the current time.

Prerequisites

You should:

- be familiar with basics of logic and digital design,
- be familiar with Verilog constructs,
- have Quartus Prime Lite installed.

Objectives

You will familiarize yourself with the following:

- 1 How to create design in Quartus Prime
- 2 How to simulate design using ModelSim
- 3 How to debug design on a board using ISSP and SignalTapII.
- 4 How to implement design on a board.

Before going forward

Before going forward with the Lab be sure that you have all files in the working folder.




Figure 1

This lab utilizes a simple digital logic design. There is only one source file with a description of the design.

The source code is in the file **C:\Intel_trn\Q_M_Debug\Lab3_1\Lab3_1.v**.

```

1  `timescale 1ns/1ns
2  module Lab3_1
3  #( parameter div_par = 25'd4) (
4      input CLK,
5      input RST,
6      input DIR,
7      output [3:0] DIG,
8      output reg [6:0] HEX
9  );
10     reg [24:0] div_cnt = 25'd1; //Clock divider
11     wire cout; //Carry out
12     reg [3:0] cnt_val; //value to count
13     reg [1:0] rst_int = 2'd0; //Synchronized reset
14     reg [3:0] Counter = 4'd0; //Counter
15     //=====
16     // Reset
17     //
18     always @(posedge CLK)
19         rst_int <= {rst_int[0], RST};
20     //=====
21     // Clock Divider
22     //
23     always @(posedge CLK, negedge rst_int[1])
24         if(!rst_int[1])
25             div_cnt <= 25'd1;
26         else
27
28
29             div_cnt <= div_cnt + 25'd1;
30
31
32     assign cout = (div_cnt == div_par);
33
34     // Counter
35     //
36     always @(posedge CLK)
37         cnt_val = (DIR)? (-4'd1):(4'd1); //value for counting
38
39     always @(posedge CLK, negedge rst_int[1])
40         if(!rst_int[1])
41             Counter <= 4'd0;
42         else
43             if (cout)
44                 begin
45                     Counter <= Counter + cnt_val;
46                     case ({DIR, Counter})
47                         5'b10000: Counter <= 4'd9;
48                         5'b01001: Counter <= 4'd0;
49                     endcase
50                 end
51     //=====
52     // Coder
53     //
54     always @(posedge CLK, negedge rst_int[1])
55         if(!rst_int[1]) HEX <= 7'b0111111;
56         else
57             case(Counter)
58                 4'b0000: HEX <= 7'b0111111; // "0"
59                 4'b0001: HEX <= 7'b0000110; // "1"
60                 4'b0010: HEX <= 7'b1011011; // "2"
61                 4'b0011: HEX <= 7'b1001111; // "3"
62                 4'b0100: HEX <= 7'b1100110; // "4"
63                 4'b0101: HEX <= 7'b1101101; // "5"
64                 4'b0110: HEX <= 7'b1111101; // "6"
65                 4'b0111: HEX <= 7'b0000111; // "7"
66                 4'b1000: HEX <= 7'b1111111; // "8"
67                 4'b1001: HEX <= 7'b1101111; // "9"
68                 default: HEX <= 7'b0111111; // "0"
69             endcase
70     //=====
71     // Constant value
72     //
73     assign DIG = 4'b1000;
74     //=====
75     endmodule

```

Figure 2

Inputs for the top-level module Lab3_1 are:

- CLK – clock signal. It is FPGA pin connected to a Clock Generator on the board. Frequency is 25MHz.
- RST – asynchronous reset signal. It is FPGA pin connected to Push Button on the board.
- DIR – direction for counting (1 – counting DOWN ; 0 – Counting UP). It is FPGA pin connected to a switch on the board.

Outputs for the top-level module Lab3_1 are:

- [6:0] HEX – 7-segment code. These are FPGA pins connected to 7-segment display on a development board.
- [3:0] DIG – constant value. These are FPGA pins connected to inputs switching on/off digits of 7-segment display.


Algorithm of the project is:

- Clock Divider (**div_cnt**) divides input clock and provides carry out (**cout**) signal.
 - The value of division (**div_par**) is a parameter.
- Counter (**cnt_val**) is

- binary-decimal counter
- synchronized by clock (**CLK**),
- enabled by carry out (**cout**) signal, coming from Clock Divider.
- the value for counting (**cnt_val**) is based on input DIR
- **Coder**
 - Codes 4-bits binary-decimal code, coming from Counter, into 7-segment code.
 - Sets constant value “1000” for DIG outputs (the leftmost digit of 7-segment indicator is on, the rest are off).

Pay attention that the code provided has some issues, which you will fix during simulation procedure.

Creating the project in Quartus Prime

- 1 Start Quartus Prime **Lite** development tool
- 2 Create a project
 - a. Working directory: **C:\Intel_trn\Q_M_Debug\Lab3_1**
 - b. Project name: **Lab3_1**
 - c. Top-Level design entity: **Lab3_1**
 - d. Project Type: Empty project
 - e. Add files from the folder C:\Intel_trn\Q_M_Debug\Lab3_1: **Lab3_1.v**
 - f. Device: **EP4CE6E22C8**
 - g. EDA tools: **NONE**
- 3 In Quartus Prime window, select **Assignment => Settings...**
- 4 In the window appeared:
 - a. Select **Compilation Process Settings**
 - i Set **Use all available processors**
 - ii Set **Use Smart Compilation**
 - iii Click **OK**
- 5 To check if the project is correct:
 - a. Select **Project navigator** window => **Hierarchy** in the drop down menu.
 - b. Select **Lab3_1** top level unit
 - c. Select **Processing** menu => **Start => Start Analysis and Synthesis** 
 - d. Be sure that you have neither Errors nor Critical Warnings.
 - i You can have one warning dealing with constant values on DIG outputs

```

▼ ⚠ 13024 Output pins are stuck at VCC or GND
    ⚠ 13410 Pin "DIG[0]" is stuck at GND
    ⚠ 13410 Pin "DIG[1]" is stuck at GND
    ⚠ 13410 Pin "DIG[2]" is stuck at GND
    ⚠ 13410 Pin "DIG[3]" is stuck at VCC
  
```

Figure 3

Simulating the design in ModelSim

Top-level file for simulation

The source code for the testbench is in the file C:\Intel_trn\Q_M_Debug\Lab3_1\tb_Lab3_1.v.

```
1  `timescale 1ns/1ns
2  module tb_Lab3_1 ();
3      reg          tb_clk;
4      reg          [5:0]  tb_mem [0:127];
5      reg          tb_dir;
6      wire         [3:0]  tb_dig;
7      wire         [6:0]  tb_hex;
8      reg          tb_reset;
9      wire         [6:0]  tb_ss;
10 initial
11 begin : clock_gen
12     tb_clk = 1'b0;
13     while (1) #10 tb_clk = ~tb_clk;
14 end
15
16 Lab3_1 #(25'd4) Lab3_1_inst (
17     .CLK      (tb_clk      ),
18     .RST      (tb_reset    ),
19     .DIR      (tb_dir      ),
20     .DIG      (tb_dig      ),
21     .HEX      (tb_hex      )
22 );
23
24 initial
25 begin: reset_gen
26     tb_reset = 1'b0;
27     #100 tb_reset = 1'b1;
28 end
29
30 initial
31 begin : control_gen
32     tb_dir = 1'b0;
33     #1100 tb_dir = 1'b1;
34     #880;
35 end
36
37 initial
38 begin : mem_reading
39     $readmemb ("ss_to_ascii.txt", tb_mem);
40     #2220 $stop;
41 end
42
43 assign tb_ss = tb_mem [tb_hex];
44
45 endmodule
```


Figure 4

This is a simple testbench in which there are:

- Clock_gen – a procedural block for clock signal (**tb_clk**) generation.
- Reset_gen – a procedural block for reset signal (**tb_reset**) generation.
- Control_gen – a procedural block for counting direction signal (**tb_dir**) generation.
- Mem_reading – a procedural block for reading data from file **ss_to_ascii.txt** into array **tb_mem**.
- Lab3_1_inst – an instantiation of entity Lab3_1.

7-segment code, coming from HEX bus outputs of Lab3_1_inst, are decoded into ASCII symbols by using Look-up table implemented as the array **tb_mem**.

File **ss_to_ascii.txt** contains address (in hexadecimal code), which are 7-segment codes, and data values (in binary code), which are ASCII codes for digital values coded in 7-segment code.

 ss_to_ascii.txt

```
@3F 110000
@06 110001
@5B 110010
@4F 110011
@66 110100
@6D 110101
@7D 110110
@07 110111
@7F 111000
@6F 111001
```

Figure 5

- 1 Start ModelSim Intel FPGA Starter Edition.








Figure 6


- 2 Change Directory.
 - a. Select **File** => **Change Directory** and change to the directory **C:\Intel_trn\Q_M_Debug\Lab3_1**.
 - b. *OR type in the Transcript window: `cd C:\Intel_trn\Q_M_Debug\Lab3_1`*
- 3 Create the working library.
 - a. Select **File** => **New** => **Library** => **work**.
 - b. *OR type in the Transcript window: `vlib work`*
- 4 Compile source files:
 - a. Select **Compile** => **Compile** => "Lab3_1.v" "tb_Lab3_1.v".
 - b. *OR type in the Transcript window: `vlog -work work "Lab3_1.v" "tb_Lab3_1.v"`*
 - c. Be sure that there are no any Errors. If there are ones you will need to correct source files.
- 5 Now you're ready to load the design into the simulator:
 - a. In the Library window, click the '+' sign next to the work library to show the files contained there and double-click **tb_Lab3_1** to load the design.
 - b. *OR type in the Transcript window: `vsim work.tb_Lab3_1`*
 - c. Be sure that there are no errors reported in the Transcript window.

Simulation and debugging

- 1 In the Objects window
 - a. Right-click in the window and select **Add to** => **Wave** => **Signals in region**
- 2 In the Wave window appeared
 - a. Right-click **tb_ss** signal and select **Radix** => **ASCII**
 - b. Right-click **tb_hex** signal and select **Radix** => **Hexadecimal**
- 3 Start simulation by clicking  icon.
 - a. *Or by typing in the Transcript window: `run -all`*
- 4 Zoom full Wave window.
- 5 Pay attention on Wave window – you will see that the project does not work properly.



- a. You can see a value 0 (ASCII), then a value 1 (ASCII) and then it seems that project stalls. It keeps value 1 (ASCII) by the end of simulation.
- 6 To find a reason of such behavior add to the Wave window some signals from Lab3_1 unit:
 - a. In the **Sim** window select **Lab3_1_inst**
 - b. In the **Object** window select: **div_cnt, cout, Counter**
 - c. Right-click and select **Add to => Wave => Selected Signals**
 - d. In the Wave window
 - i select **div_cnt** and **Counter**,
 - ii right-click and select **Radix => Unsigned**
- 7 Restart simulation:
 - a. Click  icon
 - i OR type in the transcript window: **restart**.
 - b. Click OK in the window appeared (if it is appeared.).
- 8 Start simulation by clicking  icon.
 - a. Or by typing in the Transcript window: **run -all**
- 9 Zoom in Wave window.



- 10 Pay attention on Wave window:
 - a. You can see that at the beginning the **cout** signal is correct (**div_cnt** has value 4, in accordance with the parameter value in tb_Lab3_1.v file).
 - b. Then the **div_cnt** continues to count instead of re-starting to count from 1. It seems that Clock Divider (**div_cnt**) in Lab3_1.v does not work properly. You need to check and correct **div_cnt** description in Lab3_1.v file (lines 25-33). You can use any text editor for it.

The hint:

```

21 // Clock Divider
22 //
23 always @(posedge CLK, negedge rst_int[1])
24     if(!rst_int[1] )
25         div_cnt <= 25'd1;
26     else
27         if (cout)
28             div_cnt <= 25'd1;
29         else
30             div_cnt <= div_cnt + 25'd1;
31
32 assign cout = (div_cnt == div_par);
33 //=====

```

Figure 7

- 11 In **ModelSim**: recompile Lab3_1.v

- a. Select **Library** window.
 - b. Click '+' sign next to the **work** library to show the files contained there.
 - c. Right-click **Lab3_1** and select **Recompile**.
 - d. Be sure that there are no any Errors. If there are ones you will need to correct the source file.
- 12 Restart simulation:



- a. Click icon
i OR type in the transcript window: **restart**.
 - b. Click OK in the window appeared.
- 13 Start simulation by clicking icon.
- a. Or by typing in the Transcript window: **run -all**
- 14 Zoom Full Wave window.
- 15 Pay attention on Wave window
- a. The project works properly for now
 - i On **tb_ss** bus you can see ASCII values form 0 up to 9
 - ii The Counter counts Up and Down in accordance with DIR values.
 - iii The Counter counts as Binary-Decimal counter.




Figure 8

- 16 In **Memory List** window:
- a. Double-click **tb_mem**
 - b. In the window appeared:
 - i Right-click in the address area and select **Properties**
 - ii In the window appeared select **Words per Line = 8** and click **OK**.
 - iii In **Memory Data** window you can see **tb_mem** memory array initialized by the values from **ss_ascii.txt** file (Address is in hexadecimal format. Data is in binary format).




Figure 9

- 17 Quit **ModelSim**: **File=>Quit**.

Debugging the design

Add IP components for debugging

Some IP components should be added for debugging purposes. These are:

- **Intel FPGA In-System Source & Probes** – the unit will be used for setting (sourcing) and watching (probing) internal project signals (instead of using real FPGA pins).
- **ALTPLL** – the unit will be used for multiplying input clock. Multiplied clock will be used in Signal Tap II unit as a clock source. Multiplied clock allows us to display source clock in the Signal Tap II as a waveform.

- 1 Open Quartus Prime Lite (if it was closed) and load project **Lab3_1**.
- 2 Create SP_unit using IP module **Intel FPGA In-System Source & Probes**
 - a. Find **Intel FPGA In-System Source & Probes** IP function by typing **In-System** in the Find field of IP catalog. If you do not have the IP Catalog already open, go to View menu => Utility Windows => IP Catalog to view the window.




Figure 10

- b. Double click **Intel FPGA In-System Source & Probes**.
- c. Name the file as **SP_unit** (Please make sure to name it as **SP_unit**) in the **New IP Variation** window that came up.




Figure 11

- d. Click **OK**.

- e. In the window appeared
- i Set Probe Port Width = **11** (7 bits for [6:0] HEX outputs, 4 bits for [3:0] DIG outputs).
 - ii Set Source Port Width = **2** (1 bit for RST input, 1 bit for DIR input).
 - iii Check in the check box **Use Source Clock** (Source ports will be synchronous).
 - iv Set Instance ID as **SP_**

Instance Info

☒ Automatic Instance Index Assignment

Instance Index: 0

The 'Instance ID' of this instance (optional): SP_

Probe Parameters

Probe Port Width [0..512]: 11

Source Parameters

Source Port Width [0..512]: 2

Hexadecimal initial value for the Source Port: 0

☒ Use Source Clock

☐ Use Source Clock Enable

Figure 12

- v Click **Finish**.
- vi In the window appeared click **Close**.
- vii In the window appeared click **Yes**.
- viii In the window appeared check that you set options in lined with Figure 13

Generation

Synthesis

Synthesis files are used to compile the system in a Quartus project.

Create HDL design files for synthesis: Verilog

☐ Create timing and resource estimates for third-party EDA synthesis tools.

☐ Create block symbol file (.bsf)

Simulation

The simulation model contains generated HDL files for the simulator, and may include simulation-only features.

Simulation scripts for this component will be generated in a vendor-specific sub-directory in the specified output directory.

Follow the guidance in the generated simulation scripts about how to structure your design's simulation scripts and how to use the *ip-setup-simulation* and *ip-make-simscrip*t command-line utilities to compile all of the files needed for simulating all of the IP in your design.

Create simulation model: None


Output Directory

Path: C:/Intel_trn/Q_M_Debug/Lab3_1/SP_unit

Generate Cancel

Figure 13

- ix Click **Generate**.
- x In the window appeared click **Finish**.

- f. In the window appeared (if it is appeared) there is a reminder that you need to add SP_unit.qip to the project manually. Click OK.
- g. In Quartus Prime select **Project => Add\Remove Files in Project...**
- h. In the window appeared **Find**  the **SP_unit.qip** file in the folder **C:\Intel_trn\Q_M_Debug\Lab3_1\SP_unit\synthesis**.
- i. Click **Open**.
- j. Be sure that the file is added to the project.

File Name	Type
SP_unit/synthesis/SP_unit.qip	IP Variation File (.qip)
Lab3_1.v	Verilog HDL File

Figure 14

- k. Click **OK**.
- 3 Create PLL_unit by using IP module **ALTPLL**
- a. Find **ALTPLL** IP function by typing **ALTPLL** in the **Find** field of IP catalog. If you do not have the IP Catalog already open, go to View menu => Utility Windows => IP Catalog to view the window.




Figure 15

- b. Double-click **ALTPLL**.
- c. Name the file as (*Please make sure to name it as **PLL_unit***) in the **Save IP Variation** window that came up.




Figure 16

- d. Click **OK**.

- e. In the window appeared set frequency for **inclk0** input as 25 MHz




Figure 17

- f. Go to folder Inputs/Lock and deselect all check boxes.




Figure 18

- g. Go to folder **Output Clocks** => **clk c0** and set **Clock multiplication factor** as 2 (to have output frequency 50MHz).




Figure 19

h. Go to Summary folder and select PLL_unit_inst.v file (it could be used as a prompt for instantiating the unit in a project file).




Figure 20

- i. Click **Finish**.
 - j. In the window appeared click **Yes** – PLL_unit.qip file will be added to the project automatically.
- 4 To check if the IPs are added to the design: select **Project navigator => IP Components** in the drop down menu.

Project Navigator		
	Entity	IP Component
	SP_unit	Intel FPGA In-System Sources & Probes
	PLL_unit	ALTPLL

Figure 21

It is necessary to create top-level file for debugging. The file **db_Lab3_1.v** is provided for the lab in the folder **C:\Intel_trn\Q_M_Debug\Lab3_1**.

```

1  module db_Lab3_1 (
2      (* altera_attribute = "-name IO_STANDARD \"3.3-V LVC MOS\"", chip_pin = "23" *)
3      input CLK
4  );
5      wire [6:0] HEX;
6      wire [3:0] DIG;
7      wire db_reset;
8      wire db_dir;
9      wire db_clk_high;
10
11  Lab3_1 Lab3_1_inst
12  (
13      .CLK      (CLK      ),
14      .RST      (db_reset ),
15      .DIR      (db_dir   ),
16      .DIG      (DIG      ),
17      .HEX      (HEX      )
18  );
19  SP_unit SP_unit_inst (
20      .source    ({db_reset, db_dir} ),
21      .probe     ({HEX, DIG}         ),
22      .source_clk (CLK               )
23  );
24  PLL_unit PLL_unit_inst(
25      .inclk0    (CLK               ),
26      .c0        (db_clk_high)
27  );
28  endmodule

```

Figure 22


The top-level file includes:

- Input CLK – it is FPGA input pin, connected with clock generator on the board.
 - chip_pin attribute sets the number of the pin.
 - altera_attribute sets I/O standard for the pin.
- Lab3_1_inst – instantiation of Lab3_1 entity.
- SP_unit_inst – instantiation of SP_unit entity.
- PLL_unit_inst – instantiation of PLL_unit entity.
 - Output clock (**c0**) of the **PLL_unit_inst** is connected with signal **db_clk_high**, which is not connected to any units in the top level description. The clock **db_clk_high** will be used as a clock source for Signal TapII logic analyzer, which will be added to the design later.

Pay attention that the source code has some issues; you will fix the issues during debugging procedure.


Analysis and Elaboration

- 1 Add the file **db_Lab3_1.v** to the project:
 - a. In Quartus Prime: select **Project => Add\Remove Files in Project...**
 - b. In the window appeared:

- i click  find icon,
- ii select the **db_Lab3_1.v** file (it is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1),
- iii Click **Open**
- iv Be sure that the you add the file to the project.

File Name	Type
db_Lab3_1.v	Verilog HDL File
> SP_unit/synthesis/SP_unit.qip	IP Variation File (.qip)
Lab3_1.v	Verilog HDL File
> PLL_unit.qip	IP Variation File (.qip)

Figure 23

- v Click OK.
- 2 Set the file **db_Lab3_1.v** as a top level file for the project
 - a. In the **Project Navigator** are select **Files**.
 - b. Right-click **db_Lab3_1.v** file and select **Set as a top level Entity**.
 - c. In the **Project Navigator** are select **Hierarchy** and check that **db_Lab3_1** is a top level entity for now.
- 3 Select: **Processing => Start => Start Analysis & Elaboration** 
- 4 Be sure that you have neither Errors nor Critical Warnings. If you got ones you need to check and correct IPs created.
- 5 Select: **Tools => Netlist Viewer => RTL Viewer**
- 6 Be sure that you have something like Figure 24.




Figure 24

- 7 Close **RTL_Viewer**.

Adding and setting up Signal TapII

We are going to use Signal Tap II logic analyzer for debugging. It is necessary to set up the logic analyzer.

- 1 Launch Signal Tap II: select **Tools => Signal Tap Logic Analyzer**
- 2 In the window appeared select **Setup tab**.

- 3 **Setup tab:** double click in the empty area to add some nodes for analysis.
- 4 In the window appeared:
 - a. In Filter field: select **Signal Tap: pre-synthesis**
 - b. Click List
 - c. In the **Matching Nodes** field select: CLK, DIR, RST, DIG, HEX, Counter, div_cnt
 - d. Click > sign




Figure 25

- e. Click Insert.
 - f. Click Close
- 5 **Setup tab:** right click in the empty area and select **Mnemonic Table Setup**
- 6 In the window appeared:
 - a. Select **Import Table**
 - b. Select file TABLE.stp in the folder C:\Intel_trn\Q_M_Debug\Lab3_1
 - c. Click Open
 - d. In the window appeared
 - i Select ss_udec: width = 7
 - ii Click > sign




Figure 26

- iii Click OK.
- e. In the window appeared:




Figure 27

- i You could see 7-segment code (left part of the table) and corresponding *mnemonic*, i.e. unsigned decimal values (right side of the table). The table will be used to simplify the debugging procedure.
 - ii Click OK.
- 7 **Setup tab:** right click **HEX[6..0]** bus and select **Bus Display Format => ss_udec: width = 7**
- a. The mnemonic table will be used for displaying values on the bus.
- 8 **Setup tab:**
- a. select **div_cnt** bus and **Counter** bus,


b. right-click and select **Bus Display Format => Unsigned decimal**

9 **Setup tab:** right click **Trigger Conditions** field (for signal **RST**) and select Rising Edge.

Type	Alias	Node	Data Enable	Trigger Enable	Trigger Conditions
		Lab3_1:Lab3_1 inst CLK	43	43	1 Basic AND
		Lab3_1:Lab3_1 inst DIR			
		Lab3_1:Lab3_1 inst RST			
		Lab3_1:Lab3_1 inst DIG[3..0]			Xh
		Lab3_1:Lab3_1 inst HEX[6..0]			XXh
		Lab3_1:Lab3_1 inst div cnt[24..0]			XXXXXXXXXXXXXXXXXXXXXXXXXXb
		Lab3_1:Lab3_1 inst Counter[3..0]			XXXXb

Figure 28

10 In the field **Signal Configuration** (main Signal Tap II window), in the area **Clock:**

- Click  button
- In the window appeared:
 - Click **List**.
 - Select **c0** output of the **PLL_unit**.
 - Click > sign.




Figure 29

- Click **OK**.
- The clock signal for **Signal Tap II** synchronization is assigned.

11 In the field **Signal Configuration** (main Signal Tap II window), in the area **Sample depth**, set 128.

12 Check Signal Configuration field and compare with the Figure 30.

Signal Configuration:

Clock: PLL_unit:PLL_unit_inst[c0]

Data

Sample depth: 128 RAM type: Auto

☐ Segmented: 2 64 sample segments

Nodes Allocated: ☒ Auto ☐ Manual: 43

Pipeline Factor: 0

Storage qualifier:

Type: Continuous

Input port:

Nodes Allocated: ☒ Auto ☐ Manual: 43

☒ Record data discontinuities

☐ Disable storage qualifier

Trigger

Nodes Allocated: ☒ Auto ☐ Manual: 43

Trigger flow control: Sequential

Trigger position: Pre trigger position

Trigger conditions: 1

☐ Trigger in

☐ Pin:

☒ Node:

☐ Instance:

☐ Hard Processor System (HPS) trigger out

Pattern: High

☐ Trigger out

☐ Pin:

☐ Instance:

☐ Hard Processor System (HPS) trigger in

☐ Hard Processor System (HPS) event: 0

Level: Active High

Latency delay:

Figure 30

13 In the field **Instance Manager** (main Signal Tap II window), in the area **Instance**, set name of the instance as **ST_unit**.

Instance Manager: Invalid JTAG configuration

Instance	Status	Enable	LEs: 896	Memory: 5	Small: 0/0	Medium: 2	Large: 0/0
ST_unit	Not runni...	<input checked="" type="checkbox"/>	896 cells	5504 bits	0 blocks	2 blocks	0 blocks

Figure 31

14 Save the current settings: select **File => Save As**

a. Type the name as: **Lab3_1.stp**

b. In the window appeared click **Yes** (it will add the Lab3_1.stp file to the project)

15 Close **Signal Tap II** window.

16 In Quartus Prime window: select **Assignment => Settings => Signal Tap Logic Analyzer**

a. Be sure that the check box is selected and the file Lab3_1.stp is pointed.

Signal Tap Logic Analyzer

Specify compilation options for the Signal Tap Logic Analyzer.

☒ Enable Signal Tap Logic Analyzer

Signal Tap File name: Lab3_1.stp

Figure 32

b. Click **OK**.

For the Full Compilation (for correct timing analysis) we need to have .sdc file with timing constraints. For this lab the file **db_Lab3_1.sdc** is provided. It is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1.

- 1 In Quartus Prime window: select **Assignment => Settings...=> Timing Analyzer.**
- 2 In the window appeared:
 - a. Find and Add the **db_Lab3_1.sdc** (it is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1).





Figure 33

- b. Click **OK**.
- 3 Start full compilation: **Processing => Start Compilation** 
 - 4 Be sure that you have neither Errors nor Critical Warnings.
 - a. It is acceptable to have some warnings like ones displayed on Figure 34




Figure 34

- 5 Check Compilation Report => Timing Analyzer:




Figure 35

- a. There are no any timing violations.
 - i The allowed Fmax for CLK input (clock_in) in the **Slowest Corner** is much higher than required 25 Mhz.
- b. Unconstrained Paths are JTAG inputs for SignalTap II.

Debugging the design

- 1 Connect the board to USB port of your PC.
- 2 Power Up your board.
- 3 In the Quartus Prime window select **Tools => Programmer** .
- 4 In the window appeared:
 - a. Click **Hardware Setup**.
 - b. In the window appeared:
 - i In the field **Available Hardware** double-click **USB-Blaster**.
 - ii Click **Close**.
 - c. Select (using **Add File..** button) file for configuring FPGA: **Lab3_1.sof** (It is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1\output_files)
 - i If the file already pointed in the field **File** you need to check that it is the file with the right name - **output_files/Lab3_1.sof**.
 - d. Select check box **Program/Configure**.




Figure 36

- e. Click **Start**.
- f. Close Programmer: **File => Close**.
- g. In the window appeared (if it is appeared) click **Yes**.
- 5 In the Quartus Prime window: select **Tools => In-System Source and Probes Editor**.
- 6 In the window appeared (if it is appeared) click **OK**.
- 7 In the **In-System Source and Probes Editor** window:
 - a. In the field **Hardware** select USB-Blaster. You will see that **SP_** instance (created on the previous steps) is recognized and the **Instance Manager** is **Ready to acquire**.
 - b. Adjust all settings, signal and busses as pointed on Figure 37.




Figure 37

- c. Right click **probe[10..4]** and select **Bus Display Format => Hexadecimal**.
- d. Right click **probe[3..0]** and select **Bus Display Format => Hexadecimal**.
- e. Right click **source[1..0]** and select **Bus Display Format => Hexadecimal**.
- f. Save current settings: **File=>Save as..**
 - i The name is **Lab3_1.spf**.

8 In the **In-System Source and Probes Editor** window:

- a. Click **Processing => Continuously Read Probe Data**



Index	Type	Alias	Name	Data	-16	-12	-8	-4	0
P[10..4]		[6:0]HEX	probe[10..4]	3Fh			3Fh		
P[3..0]		[3:0]DIG	probe[3..0]	8h			8h		
S[1..0]			source[1..0]	0h			0h		
S1		Reset	source[1]	0					
S0		DIR	source[0]	0					

Figure 38

You will see that:

- HEX bus outputs have a constant value 3F (it is 7-segment code for 0).
 - It is right while Reset signal is 0 (Active).
- DIG bus outputs have a constant value 8 .
 - It is right: the left most digit of 7-segment indicator is selected.

- b. Click in **Data** field of **Reset** signal (value from 0 will be changer to 1 and it will be displayed in red).



- c. Click **Processing => Write Source Data**




Figure 39

You will see that HEX bus outputs values are not in the expected sequence for the 7-segments codes. The expected sequence: when DIR = 0 (counting UP): 3F; 06; 5B; 4F; 66; 6D; 7D; 07; 7F; 6F ...

- d. Click **Processing => Stop Continuously Reade Probe data** .
- e. Click in **Data** field of **Reset** signal (value from 1 will be changer to 0 and it will be displayed in red).

- f. Click **Processing => Write Source Data** .
- g. **Keep the In-System Source and Probes Editor window opened!**

The hint: to understand a reason of the wrong behavior you need to check the design with Signal Tap II.

- 9 In the Quartus Prime window: select **Tools => Signal Tap Logic Analyzer**
- 10 In the **SignalTap II** window:
 - a. In the field **Hardware** select USB-Blaster. You will see that SignalTapII instance ST_unit is recognized and the **Instance Manager** is **Ready to acquire**.
- 11 In the **SignalTap II** window: select **Processing => Run Analysis**
 - a. Signal Tap II will wait a **rising** edge on RST signal.




Figure 40

- 12 In the **In-System Source and Probes Editor** window:
 - a. Click in **Data** field of **Reset** signal (value from 0 will be changer to 1 and it will be displayed in red).
 - b. Click **Processing => Write Source Data** .
- 13 In the **SignalTap II** window: you will see in the **Data** tab the captured data.




Figure 41

You can see that:

- Code Sequence (7-segment codes) on the HEX bus outputs (*The values are displayed in digital format in accordance with Mnemonic table*) is correct (*DIR = 0, i.e. Counting UP*) and is aligned with the outputs of the Counter bus.
- It seems that there are some issues with Clock divider:
 - It is expected that Counter bus outputs and Code Sequence (7-segment codes) on the HEX bus outputs change after 25 000 000 periods of input clock (CLK).
 - In the waveform, we can see that those change after 4 periods of input clock (CLK).

You need to check and correct the source files of your project.

The hint:


We already debugged **Lab3_1.v** source code during the simulation stage. Therefore, we can assume that a problem is in **db_Lab3_1.v** code. Pay attention on a division factor for the clock divider, which is a parameter in Lab3_1.v and correct db_Lab3_1.v code.

```
12 Lab3_1 #(25'd25000000) Lab3_1_inst
13 (
14     .CLK      (CLK      ),
15     .RST      (db_reset  ),
16     .DIR      (db_dir    ),
17     .DIG      (DIG       ),
18     .HEX      (HEX       )
19 );
```

Figure 42

- 14 Correct and save source code db_Lab3_1.v.
- 15 In Quartus Prime window: start full compilation: **Processing => Start Compilation.**
 - a. Click **Yes** in all windows appeared.
- 16 Be sure that you have neither Errors nor Critical Warnings.
- 17 Check **Compilation Report => Timing Analyzer.**
 - a. There are no any timing violations.
- 18 Open **Programmer** and reconfigure the device (if you wish you can do reconfiguration just from In-System Source and Probes Editor window or from Signal Tap II window).
- 19 Close Programmer.
- 20 Open **In-System Source and Probes Editor** window (re-initialize it if it is necessary by clicking **Scan Chain**)


- a. Set Reset = 0 and DIR = 0.

- b. Click **Processing => Write Source Data** .

- c. Click **Processing => Continuously Read Probe Data** .

You will see 3F on HEX output. It is right result.

- d. Set Reset = 1 .

- e. Click **Processing => Write Source Data** .

You will see the right code sequence on HEX bus outputs. The expected sequence:

- For DIR = 0 (Counting UP): 3F; 06; 5B; 4F; 66; 6D; 7D; 07; 7F; 6F ...

Index	Type	Alias	Name	Data	-16	-14	-12	-10	-8	-6	-4
P[10..4]		[6:0] HEX	probe[10..4]	4Fh	3Fh	06h	5Bh	4Fh	66h	6Dh	7Dh
P[3..0]		[3:0] DIG	probe[3..0]	8h						8h	
S[1..0]			source[1..0]	2h	0h					2h	
S1		Reset	source[1]	1							
S0		DIR	source[0]	0							

Figure 43

f. Set DIR = 1.

g. Click **Processing => Write Source Data**

You will see the right code sequence on HEX output. The expected sequence:

- For DIR = 1 (Counting Down): 3F; 6F; 7F; 07; 7D; 7D; 6D; 66; 4F; 5B; 06; ...

Index	Type	Alias	Name	Data	-16	-14	-12	-10	-8	-6	-4	-2	0
P[10..4]		[6:0] HEX	probe[10..4]	7Fh	6Fh	3Fh	06h	3Fh	6Fh	7Fh	07h	7Dh	6Dh
P[3..0]		DIG	probe[3..0]	8h							8h		
S[1..0]			source[1..0]	3h	2h						3h		
S1		Reset	source[1]	1									
S0		DIR	source[0]	1									

h. Click **Processing => Stop Continuously Read Probe data**

i. Set DIR = 0.

j. Click **Processing => Write Source Data**

21 Open **Signal Tap II** window (re-initialize it if it is necessary by clicking **Scan Chain**).

22 In the **Signal Tap II** window:

- Tab Setup:** Change Trigger Conditions filed of RST signal to Don't Care.
 - Right click the field and select .
- Tab Setup:** Change Trigger Conditions for **div_cnt** bus to 25000000.
 - Right click in the field **Trigger Condition** for **div_cnt** bus => Insert Value.
 - Type 25000000.
 - Click Ok.

Node			Data Enable	Trigger Enable	Trigger Conditions
Type	Alias	Name	43	43	1 Basic AND
		Lab3 1:Lab3 1 inst CLK	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	
		Lab3 1:Lab3 1 inst DIR	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	
		Lab3 1:Lab3 1 inst RST	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	
		Lab3 1:Lab3 1 inst DIG[3..0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	Xh
		Lab3 1:Lab3 1 inst HEX[6..0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXh
		Lab3 1:Lab3 1 inst Counter[3..0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	XXXXb
		Lab3 1:Lab3 1 inst div cnt[24..0]	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	25000000

Figure 44

23 In the **Signal Tap II** window:

a. Click **Processing => Run Analysis**

b. Zoom in to see waveform like on Figure 45.

You could see that the signal **div_cnt** equal the value 25000000 and then start to count from 1.




Figure 45

- 24 In the **Signal Tap II** window:
 - a. **Tab Setup:** Change Trigger Conditions field of HEX bus to Either Edge (capital E).
 - i Right click in the field **Trigger Condition** for **HEX** bus => Either Edge.
 - b. **Tab Setup:** Change Trigger Conditions field of div_cnt bus to Don't care (letter X).
 - i Right click in the field **Trigger Condition** for **div_cnt** bus => Don't care.
 - c. **Tab Setup:** Set **Segmented Acquisition Trigger** field (field which is just under the Trigger Conditions name) as **Basic OR**.
 - d. **Signal Configuration** field: Check **Segmented** check box.
 - e. **Signal Configuration** field: Select value for Segmented as **16 8 sample segments**.
 - f. **Signal Configuration** field: in **Trigger Position** area set **Center Trigger Position**.




Figure 46

- g. Save file with current settings.
- 25 In the **Signal Tap II** window: start full compilation selecting **Processing => Start Compilation**
- 26 Be sure that you have neither Errors nor Critical Warnings.
- 27 Check **Compilation Report => Timing Analyzer**.
 - b. There are no any timing violations.
- 28 In Quartus Prime window: open Programmer and reconfigure the device (you can do reconfiguration just from In-System Source and Probes Editor window and Signal Tap II windows too.).
- 29 Close Programmer.
- 30 Open **In-System Source and Probes Editor** window (re-initialize it if it is necessary by clicking **Scan Chain**).
 - a. Set Reset = 1 and DIR =0.



b. Click **Processing => Write Source Data**.

31 Open **Signal Tap II** window (re-initialize it if it is necessary by clicking **Scan Chain**).


32 In the **Signal Tap II** window:



a. Click **Processing => Run Analysis**.

b. You will need to wait near 16s for getting data: Signal Tap II need to wait 16 values on the HEX bus. The values are updated one time per second.

You will see the right sequence on the HEX bus output for DIR = 0 – i.e. counting UP (values on the bus are from the Mnemonic Table).



33 In In-System Source and Probes Editor window

a. Set DIR = 1.



b. Click **Processing => Write Source Data**.

34 In Open Signal Tap II window.



a. Click **Processing => Run Analysis**.

b. You will need to wait near 16s for getting data: Signal Tap II need to wait 16 values on the HEX bus. The values are updated one time per second.

You will see the right sequence on the HEX bus output for DIR = 1 – i.e. counting DOWN (values on the bus are from the Mnemonic Table).




Figure 47

35 Save and close **Signal Tap II** widow.

36 Save and close **In-System Source and Probes Editor** widow.

Implementing the design

Top-level file for implementation


Source file for implementation stage should be based on db_Lab3_1.v file with some modifications implemented:

- All inputs and outputs of the project should be added and associated with FPGA inputs
- All additional components, used for debugging purposes, should be removed.

The source file impl_Lab3_1.v is provided for the lab. It is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1.

```
1  module impl_Lab3_1 (
2      (* altera_attribute = "-name IO_STANDARD \"3.3-V LVC MOS\"", chip_pin = "23" *)
3      input CLK,
4      (* altera_attribute = "-name IO_STANDARD \"2.5 V\"", chip_pin = "64" *)
5      input RST,
6      (* altera_attribute = "-name IO_STANDARD \"3.3-V LVC MOS\"", chip_pin = "88" *)
7      input DIR,
8      (* altera_attribute = "-name IO_STANDARD \"3.3-V LVC MOS\"", chip_pin = "84, 76, 85, 77, 86, 133, 87" *)
9      output [6:0] HEX,
10     (* altera_attribute = "-name IO_STANDARD \"3.3-V LVC MOS\"", chip_pin = "73, 74, 80, 83" *)
11     output [3:0] DIG
12 );
13
14 Lab3_1 #(25'd25000000) Lab3_1_inst
15 (
16     .CLK      (CLK      ),
17     .RST      (RST      ),
18     .DIR      (DIR      ),
19     .DIG      (DIG      ),
20     .HEX      (HEX      )
21 );
22
23 endmodule
```

Figure 48

1. Add the file **impl_Lab3_1.v** to the project:
 - a. In Quartus Prime select **Project => Add\Remove Files in Project...**
 - b. In the window appeared **Find**  the **impl_Lab3_1.v** file in the folder **C:\Intel_trn\Q_M_Debug\Lab3_1**
 - c. Click **Open**
 - d. Click **OK**.
2. Set the file **impl_Lab3_1.v** as a top level file for the project
 - a. Select: **Project Navigator => Files**
 - b. Right-click **impl_Lab3_1.v** file and select **Set as a top level Entity**
 - c. In the **Project Navigator** select **Hierarchy** and check that **impl_Lab3_1** is a top level entity.
3. Select: **Processing => Start => Start Analysis & Elaboration**
4. Select: **Tools => Netlist Viewer => RTL Viewer**
5. Be sure that you got something like Figure 49.




Figure 49

Full compilation

For the Full Compilation (for correct timing analysis) we need to have .sdc file with timing constraints. For this lab the file **impl_Lab3_1.sdc** is provided. It is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1.

- 1 In Quartus Prime window: select **Assignment => Settings...**
- 2 In the window appeared:
 - a. Select Timing Analyzer
 - i Select **db_Lab3_1.sdc** file and click Remove
 - ii Find and Add the **impl_Lab3_1.sdc** (it is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1)

File Name	Type
impl_Lab3_1.sdc	Synopsys Design Constraints File

Figure 50

- b. Click **Apply**.
 - c. Select **Signal Tap Logic Analyzer**.
 - i Clear check box **Enable Signal Tap Logic Analyzer**.
 - d. Click **OK**.
- 3 In Quartus Prime window: start full compilation selecting **Processing => Start Compilation**
- 4 Be sure that you have neither Errors nor Critical Warnings after Full compilation.
 - a. It is acceptable to have some warnings like ones displayed on Figure 51.




Figure 51

- 5 Check Compilation Report => Timing Analyzer:

Table of Contents				Slow 1200mV 85C Model Fmax Summary			
•	Flow Messages						
•	Flow Suppressed Messages						
>	Assembler						
▼	Timing Analyzer						
	Summary						
	Parallel Compilation						
	SDC File List						
	Clocks						
▼	Slow 1200mV 85C Model						
	Fmax Summary						
	Timing Closure Recommendations						
	Setup Summary						
	Hold Summary						
	Recovery Summary						
	Removal Summary						
	Minimum Pulse Width Summary						
>	Worst-Case Timing Paths						
	Metastability Summary						
>	Slow 1200mV 0C Model						
>	Fast 1200mV 0C Model						
	Multicorner Timing Analysis Summary						
>	Advanced I/O Timing						
>	Clock Transfers						
	Report TCCS						
	Report RSKM						
>	Unconstrained Paths						

<<Filter>>			
	Fmax	Restricted Fmax	
1	83.06 MHz	83.06 MHz	clock_in

Figure 52

- a. There are no any timing violations.
 - i The allowed Fmax for CLK input (clock_in) in the **Slowest Corner** is much higher than required 25 Mhz.

Programming and verifying the board

- 1 Connect the board to USB port of your PC.
- 2 Power Up your board.
- 3 In the Quartus Prime window select **Tools => Programmer** .
- 4 In the window appeared:
 - a. Click on **Hardware Setup**
 - b. In the window appeared:
 - i in **Available Hardware** field double-click **USB-Blaster**
 - ii Click Close
 - c. Select (using **Add File..** button) file for configuring FPGA: **Lab3_1.sof** (It is in the folder C:\Intel_trn\Q_M_Debug\Lab3_1\output_files)
 - i If the file already pointed in the field **File** you need to check that it is the file with the right name - **output_files/Lab3_1.sof**
 - d. Select check box **Program/Configure**




Figure 53

- e. Click **Start**.
 - f. Close Programmer: File => Close.
- 5 On the Board:
- a. Set sw [7] = 0
 - i Check that 7-segment indicator displays 0, 1, ...9, 0...
 - b. Set sw [7] = 1
 - i Check that 7-segment indicator displays 9, 8, ... 0, 9...
 - c. Push button PBA
 - i Check that 7-segment indicator displays 0

Conclusions

With this lab you learnt how to:

- create design in Quartus Prime
- simulate and debug source code using ModelSim
- debug design on board using ISSP and SignalTapII.
- Implement and check the design on the board.