Simulation, debugging and implementing the design with ModelSim and Quartus Prime Lite

самостоятельная работа с использованием SystemVerilog

ИНДИВИДУАЛЬНЫЕ ЗАДАНИЯ

В РАЗДЕЛЕ

ЗАДАНИЕ

Contents

Задание.

Лабораторная работа выполняется по индивидуальным заданиям. Теория и приведенные ниже (на английском языке) примеры могут быть использованы как образцы для составления собственных исходных кодов проектируемых модулей и тестов. В работе следует использовать конструкции SystemVerilog.

Программа работы

- Разработать описание LFSR по индивидуальному заданию (номер задания = номеру в списке группы)
- Разработать тест для проверки LFSR и провести моделирование
- По результатам моделирования в ModelSim **необходимо доказать**, что период повторения равен 2^N-1, где N – максимальная степень полинома из задания (ОБРАТИТЕ ВНИМАНИЕ: в некоторых вариантах задания, теоретически, может быть ошибка, тогда этот период будет меньше 2^N-1. Если у Вас так получилось, то надо, прежде всего, проверить правильность своего описания, и только после этого обсудить это с преподавателем).
- Разработать модуль для построения гистограммы
- Разработать тест для проверки модуля построения гистограммы и провести моделирование
- По результатам моделирования в ModelSim необходимо доказать, что для входных данных, поступающих со счетчика (шаг счета = номеру студента в группе) получена правильная гистограмма в модуле памяти.
- Разработать модуль верхнего уровня lab_MS_SV3, содержащий LFSR и модуль построения гистограммы.
- Разработать тест для проверки модуля верхнего уровня иерархии (tb_lab_MS_SV3).
- По результатам моделирования необходимо доказать, что период повторения LSFR равен 2^N-1, где N – максимальная степень полинома из задания и модуль построения гистограммы создает правильную гистограмму.
- Разработать модуль верхнего уровня для отладки db_lab_MS_SV3, содержащий: модуль lab_MS_SV3; модуль SP_unit (модуль, обеспечивающий возможность задания входных управляющих сигналов без использования кнопок на плате). Модуль должен содержать подключение только к тактовому сигналу на плате.
- Настроить логический анализатор для проведения исследования и отладки реализуемого на плате db_lab_MS_SV3.
- Провести анализ работы db_lab_MS_SV3 и доказать (зафиксировав результаты снимками экрана), что:
    - Модуль управляется входными сигналами RST и ENA
    - Правильно формируется гистограмма
    - Формируемые псевдослучайные данные отображаются аналогично изображению на Figure 17.

Дополнительное задание

- Изменение модуля histogram_unit
  - В модуле histogram_unit (см. Figure 7) замените память mem_in, созданную как массив, на параметризированный (IP) блок RAM:1_PORT
    - 10 бит – разрядность слова, 1024 слова, Read-During Write – OLD DATA;
    - Включите Allow in-System-Memory-Content Editor и задайте Instance ID = RAM0
  - Осуществите моделирование в пакете ModelSim (не забудьте подключить библиотеку altera_mf_ver) и убедитесь в правильности работы модуля.
- В модуле db_lab_MS_SV3 настройте в логическом анализаторе 16 сегментов (по 16 отсчетов, положение – в центре) условие захвата – достижение генератором значения, равного Вашему номеру в списке группы.
- Запустите ISSP, убедитесь, что RST=1 и ENA= 1;
- Запустите IMCE: считайте данные из модуля памяти с Instance ID = RAM0. Зафиксируйте их и объясните полученные результаты.
- Откройте окно Логического анализатора SignalTapII и запустите захват данных
- Откройте окно ISSP, установите RST=0
- Откройте окно IMCE: считайте данные из модуля памяти с Instance ID = RAM0. Зафиксируйте их и объясните полученные результаты.
- Откройте окно Логического анализатора SignalTapII, зафиксируйте и приведите в отчете временные диаграммы, убедитесь в том, что:
  - временная диаграмма похожа на приведенную ниже

  - формируемые генератором данные, повторяются во всех 16-х сегментах.
  - объясните почему на выходе mem_out мы видим числа 0, 1, 2, …15.
    - если увеличить количество сегментов для захвата данных, то до какого занчения будут увеличиваться эти числа?
      - Как это объяснить?
      - Докажите это с использованием SignalTapII

Содержание отчета

- Отчет должен быть оформлен по правилам принятым в Высшей Школе.
- Отчет должен содержать все этапы работы, все созданные исходные коды, необходимые снимки экрана. Все рисунки и полученные на них результаты должны быть прокомментированы.

Номер задания = номеру студента в списке группы	**Полином для реализации** **(регистр 8 или 7 разрядный, в зависимости от максимальной степени в полиноме)**	**Тип реализации** **Тип логического элемента в обратной связи.**
1.	$x^8 + x^7 + x^6 + x^5 + x^4 + x^3 + 1$	Галуа XOR
2.	$x^8 + x^7 + x^6 + x^5 + x^4 + x^2 + 1$	Фибоначчи XOR
3.	$x^8 + x^7 + x^6 + x^5 + x^4 + x + 1$	Фибоначчи XOR
4.	$x^8 + x^7 + x^6 + x^5 + x^2 + x + 1$	Фибоначчи XNOR
5.	$x^8 + x^7 + x^6 + x^4 + x^3 + x^2 + 1$	Галуа XOR
6.	$x^8 + x^7 + x^6 + x^4 + x^2 + x + 1$	Фибоначчи XOR
7.	$x^8 + x^7 + x^6 + x^3 + x^2 + 1$	Фибоначчи XOR
8.	$x^7 + x^6 + x^5 + x^4 + x^3 + x^2 + 1$	Фибоначчи XNOR
9.	$x^8 + x^4 + x^3 + x + 1$	Галуа XOR
10.	$x^8 + x^4 + x^3 + x^2 + 1$	Фибоначчи XOR
11.	$x^8 + x^5 + x^3 + x + 1$	Фибоначчи XOR
12.	$x^8 + x^5 + x^3 + x^2 + 1$	Фибоначчи XNOR
13.	$x^8 + x^5 + x^4 + x^3 + 1$	Галуа XOR

14.	$x^8 + x^6 + x^3 + x^2 + 1$	Фибоначчи XOR
15.	$x^7 + x^5 + x^2 + x + 1$	Фибоначчи XOR
16.	$x^7 + x^3 + x^3 + x + 1$	Фибоначчи XNOR
17.	$x^7 + x^5 + x^4 + x^3 + 1$	Галуа XOR
18.	$x^7 + x^6 + x^3 + x + 1$	Фибоначчи XOR
19.	$x^7 + x^6 + x^4 + x + 1$	Фибоначчи XOR
20.	$x^7 + x^6 + x^5 + x^2 + 1$	Фибоначчи XNOR
21.	$x^7 + x^6 + x^5 + x^4 + x^2 + x + 1$	Галуа XOR
22.	$x^7 + x^6 + x^5 + x^3 + x^2 + 1$	Фибоначчи XOR
23.	$x^7 + x^4 + 1$	Фибоначчи XOR
24.	$x^7 + x^3 + 1$	Фибоначчи XNOR
25.	$x^7 + x + 1$	Галуа XOR
26.	$x^7 + x^3 + x^2 + x + 1$	Фибоначчи XOR
27.	$x^7 + x^6 + 1$	Фибоначчи XOR
28.	$x^7 + x^6 + x^5 + x^4 + 1$	Фибоначчи XNOR
29.	$x^7 + x^3 + x^3 + x + 1$	Галуа XOR
30.	$x^7 + x^6 + x^5 + x^3 + x^2 + 1$	Фибоначчи XOR

Create the project in Quartus Prime

1 Start Quartus Prime **Lite** development tool

2 Create a project

a. Working directory: **C:\Intel_trn\Q_MS_SV\lab_MS_SV3**
b. Project name: **lab_MS_SV3**
c. Top-Level design entity: **lab_MS_SV3**
d. Project Type: Empty project
e. Add files: No files

- Device:
  - **Cyclone IV E**
    - **EP4CE6E22C8 –** для платы **MiniDilab-CIV**
  - **MAX10**
    - **10M50DAF484C6GES** – для платы **MAX10 NEEK**

f. EDA tools: ModelSim ASE => SystemVerilog

Developing LFSR unit

Introduction to theory

Linear-feedback shift register (LFSR) is a shift register whose input bit is a linear function of its previous state. The initial value of the LFSR is called the seed. The seed could be any value except zero. Applications of LFSRs include generating pseudo-random numbers, pseudo-noise sequences, fast digital counters, and whitening sequences. The following table lists some examples of maximal-length polynomials for shift-register lengths up to 16.

Биты, n	Примитивный многочлен	Период, $2^n - 1$	Число примитивных многочленов
2	$x^2 + x + 1$	3	1
3	$x^3 + x^2 + 1$	7	2
4	$x^4 + x^3 + 1$	15	2
5	$x^5 + x^3 + 1$	31	6
6	$x^6 + x^5 + 1$	63	6
7	$x^7 + x^6 + 1$	127	18
8	$x^8 + x^6 + x^5 + x^4 + 1$	255	16
9	$x^9 + x^5 + 1$	511	48
10	$x^{10} + x^7 + 1$	1023	60
11	$x^{11} + x^9 + 1$	2047	176
12	$x^{12} + x^{11} + x^{10} + x^4 + 1$	4095	144
13	$x^{13} + x^{12} + x^{11} + x^8 + 1$	8191	630
14	$x^{14} + x^{13} + x^{12} + x^2 + 1$	16383	756
15	$x^{15} + x^{14} + 1$	32767	1800
16	$x^{16} + x^{14} + x^{13} + x^{11} + 1$	65535	2048

Figure 1

Let's explore 16-bit LFSR: $X^{16}+X^{14}+X^{13}+X^{11}+1$ in Fibonacci and Galois configurations. The bit positions that affect the next state are called the ***taps***.

Fibonacci configuration.



Figure 2

In the diagram the taps are [16,14,13,11]. The rightmost bit of the LFSR is called the output bit. The taps are XOR'd sequentially with the output bit and then fed back into the leftmost bit.
As an alternative to the XOR-based feedback in an LFSR, one can also use XNOR:

- a state with all zeroes is illegal when using XOR feedback.
- a state with all ones is illegal when using an XNOR feedback.

This states are considered illegal because the counter would remain "locked-up" in this state.
This LFSR configuration is also known as standard, external XOR (XNOR) gates.
The alternative is Galois configuration.

It is necessary to ensure that the LFSR never enters to the all-zeros state(XOR) or to the all-ones (XNOR) state, for example by presetting it at start-up to any other state in the sequence.

Galois configuration

$X^{16}+X^{14}+X^{13}+X^{11}+1$ In the Galois configuration, the taps are XORed with the output bit (bit 1) before they are stored in the next position.



Figure 3

The Galois register shown has the same output stream as the Fibonacci register in the above section. A time offset exists between the streams, so a different start point will be needed to get the same output each cycle.
Galois LFSRs do not concatenate every tap to produce the new input (the XORing is done within the LFSR, and no XOR gates are run in serial, therefore the propagation times are reduced to that of one XOR rather than a whole chain), thus it is possible for each tap to be computed in parallel, increasing the speed of execution.

1 On the Figure 4 you can find an example of LFSR: $X^{10}+X^7+1$ in Fibonacci configuration with XORing.

```
1   `timescale 1ns/1ns
2   module LFSR_10_7_1_F
3       (input bit CLK, input bit RST, input bit ENA, output bit [10:1] LFSR_out);
4
5   always_ff @(posedge CLK, posedge RST)
6   if (RST)   LFSR_out <= 10'd1;
7   else if (ENA)
8           if (LFSR_out == '0)
9               LFSR_out <= 10'd1;
10          else
11              LFSR_out <= {LFSR_out[9:1],LFSR_out[10]^LFSR_out[7]};
12  endmodule
```

Figure 4

2 On the Figure 5 you can find an example of LFSR: $X^{10}+X^7+1$ in Galois configuration with XORing.

```
1   `timescale 1ns/1ns
2   module LFSR_10_7_1_G
3       (input bit CLK, input bit RST, input bit ENA, output bit [10:1] LFSR_out);
4
5   always_ff @(posedge CLK, posedge RST)
6   if (RST)   LFSR_out <= 10'd1;
7   else if (ENA)
8           if (LFSR_out == '0)
9               LFSR_out <= 10'd1;
10          else
11              LFSR_out <= {LFSR_out[1],LFSR_out[10:9],LFSR_out[1]^LFSR_out[8],LFSR_out[7:2]};
12  endmodule
```

Figure 5

The both implementations have the same inputs and output.

- Inputs are
  - CLK – clock input.
  - RST – asynchronous reset (when = 1'b1) to the state 10'b0000000001. It is a seed for the LFSR.
  - ENA – enable input (enable to work when is equal to 1'b1).
- Output
  - LFSR_out – 10 bits output of the LFSR.
- Pay attention on:
  - In line 8 the ***if*** is used to ensure that LFSR never enters an all-zeros state.

An example of a source code for the testbench is on Figure 6.

```
1   `timescale 1ns/1ns
2   module tb_LFSR_10_7_1();
3     bit         CLK;
4     bit [10:1]  LFSR_out_F ;
5     bit [10:1]  LFSR_out_F1;
6     bit [10:1]  LFSR_out_G;
7     bit [10:1]  LFSR_out_G1;
8     bit [10:1]  CNT_int=10'd1;
9     bit RST;
10    bit ENA ;
11    bit LFSR_CYCLE;
12  LFSR_10_7_1_G LFSR_10_7_1_G_inst (
13  .LFSR_out (LFSR_out_G),
14  .CLK      (CLK),
15  .ENA      (ENA),
16  .RST      (RST)
17  );
18  LFSR_10_7_1_F LFSR_10_7_1_F_inst (
19  .LFSR_out (LFSR_out_F),
20  .CLK      (CLK),
21  .ENA      (ENA),
22  .RST      (RST)
23  );
24  always  #10 CLK = ~ CLK;
25  initial
26  begin
27    RST = '1;
28    #15;
29    RST = '0;
30    #25;
31    ENA = '1;
32    forever begin
33      @(negedge CLK);
34      if (CNT_int=='0) begin
35        LFSR_out_G1 = LFSR_out_G;  //fixing the first data to check Galois
36       //LFSR_out_F1 = LFSR_out_F;  //fixing the first data to check Fibonacci
37        end
38      else
39        if (LFSR_out_G1 == LFSR_out_G)   //to check Galois
40        //if (LFSR_out_F1 == LFSR_out)    //to check Fibonacci
41        begin
42          LFSR_CYCLE='1;
43          break;
44        end
45      CNT_int++;
46    end
47    #100;
48    $stop;
49  end
50  endmodule
```

Figure 6

This simple testbench allows you to check Galois and Fibonacci configuration and be sure that $X^{10}+X^7+1$ is maximal-length polynomial – by the end of simulation the value of CNT_int should be 1023.

Developing the unit

The histogram detection unit allows you to build a histogram for the input data stream. An example of a source code for such unit is on Figure 7.

```
1   `timescale 1ns/1ns
2   module histogram_unit(
3       input bit           CLK,
4       input bit [9:0]     d_in,
5       input bit           RST,
6       input bit           ENA,
7       output bit [9:0]    mem_out
8   );
9   bit [9:0] mem_arr [0:1023];// an array for the histogram
10
11  bit [9:0] mem_in;
12  bit [9:0] adr_in, adr_clear;
13
14  initial
15  begin:initial_val //initial values for the array
16      for (int i=0; i<1024; i++) mem_arr [i] =0;
17  end:initial_val
18
19  assign mem_in   = (RST)? '0         : ((ENA)? (mem_out + 10'd1): mem_out);
20  assign adr_in   = (RST)? adr_clear  : d_in;
21
22  always @(posedge CLK)
23  begin:building_histogram
24      mem_arr [adr_in]    <= mem_in;
25      mem_out             <= mem_arr [adr_in];
26  end:building_histogram
27
28  always_ff @(posedge CLK, negedge RST)
29  begin:clearing_array
30      if (~RST) adr_clear <= '0;
31      else adr_clear <=adr_clear+1'b1;
32  end:clearing_array
33  endmodule
```

Figure 7

An example of a source code for the testbench is on Figure 8.

```
1   `timescale 1ns/1ns
2   module tb_histogram_unit();
3   bit [9:0] d_in;
4   bit       CLK;
5   bit       ENA;
6   bit       RST;
7   bit [9:0] mem_out;
8
9   always  #10 CLK = ~ CLK;
10
11  histogram_unit histogram_unit_inst(
12  .CLK      (CLK),
13  .d_in     (d_in),
14  .RST      (RST),
15  .ENA      (ENA),
16  .mem_out  (mem_out)
17  );
18
19  initial
20  begin
21    ENA = '1;
22    for (int i = 0; i < 4103; i++)
23      begin
24        @(negedge CLK)
25        d_in = d_in + 10'd2;
26      end
27    $stop;
28  end
29  endmodule
```

Figure 8

In this simple testbench an input stream for the histogram_unit is an output of a counter with counting step 2.

A result of simulation proving the right behavior of histogram unit is the content of the memory array (mem_arr). The memory array (mem_arr) should contains for each even address equal values 8. An example is on Figure 9.


Memory Data - /tb_histogram_unit/histogram_unit_inst/mem_arr - Default
Goto:
00000000 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
00000011 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
00000022 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
00000033 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
00000044 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
00000055 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
00000066 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
00000077 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
00000088 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
00000099 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
000000aa 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
000000bb 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
000000cc 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8
000000dd 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0
000000ee 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8 0 8

Figure 9

Developing the unit

The top level unit for the Lab contains LFSR unit and Histogram unit connected together. An example of source code is on Figure 10.

```
1   `timescale 1ns/1ns
2
3   module lab_MS_SV3(
4       input   bit     CLK,
5       input   bit     RST,
6       input   bit     ENA,
7       output  bit [9:0] mem_out);
8
9   bit [9:0] LFSR_out;
10  bit [9:0] d_in;
11
12  assign d_in = LFSR_out;
13
14  LFSR_10_7_1_G LFSR_10_7_1_G_inst (
15  .CLK        (CLK),
16  .RST        (RST),
17  .ENA        (ENA),
18  .LFSR_out   (LFSR_out)
19  );
20
21  histogram_unit histogram_unit_inst(
22  .CLK    (CLK),
23  .d_in   (d_in),
24  .RST    (RST),
25  .ENA    (ENA),
26  .mem_out(mem_out)
27  );
28
29  endmodule
```

Figure 10

In the top level unit an input stream for the histogram_unit is an output of a LFSR unit (d_in = LFSR_out).

An example of a source code for the testbench is on Figure 11.

```
1   `timescale 1ns/1ns
2   module tb_lab_MS_SV3();
3       bit CLK;
4       bit RST;
5       bit ENA;
6       bit [9:0] mem_out;
7   lab_MS_SV3 U1 (
8   .CLK        (CLK),
9   .RST        (RST),
10  .ENA        (ENA),
11  .mem_out    (mem_out)
12  );
13  always  #10 CLK = ~CLK;
14
15  initial
16  begin
17      RST = '1;
18      #5;
19      RST = '0;
20      ENA = '1;
21      repeat(4092) @(negedge CLK);
22      $stop;
23  end
24  endmodule
```

Figure 11

A result of simulation proving the right behavior of the top level unit is the content of the memory array (mem_arr). The memory array (mem_arr) should contains for each address equal values 2 (except address 0, which should be 1). An example is on Figure 12.



Figure 12

An example of a source code for debugging the unit on MiniDilbCIV board is on Figure 13.

```
module db_lab6_1
(
    (* altera_attribute = "-name IO_STANDARD \"3.3-V LVCMOS\"", chip_pin = "23"  *)
    input bit CLK
);

bit RST = 1'b1;
bit ENA = 1'b1;
bit [9:0] mem_out ;

lab6_1 lab6_1_inst (.*);
SP_unit SP_unit_inst (
    .source     ({RST, ENA} ),
    .source_clk (CLK        )
);
endmodule
```

Figure 13

For debugging purposes you need to add **Intel FPGA In-System Source & Probes** – the unit will be used for setting (sourcing) signals **RST** and **ENA** (instead of using the real FPGA pins).

You can find **Intel FPGA In-System Source & Probes** IP function by typing **In-System** in the Find field of IP catalog. If you do not have the IP Catalog already open, go to View menu => Utility Windows => IP Catalog to view the window. Name the file as **SP_unit**. Recommended settings for the unit is on Figure 14.



Figure 14

To be sure that you have the right top level project for debugging:

- Set db_lab6_1.sv file as a top level file in Quartus Prime
- Start Analyze and Elaborate procedure.
- Open RTL viewer – you should see something like structure on Figure 15.



Figure 15

For debugging purposes you need to set-up and add to the project a Signal Tap Logic Analyzer. As an example for setting it up is a Figure 16.





Figure 16

By using **In system Source and Probe** and Signal Tap Logic Analyzer be sure that the project works correctly. An example is on Figure 17. You can see that after setting RST to 0, Signal Tap Logic Analyzer got and displayed results.



Figure 17